


PIECEWISE_LINEAR_FM

* dsauersanjose@aol.com 7/10/08

* www.idea2ic.com

* PM Signal



```
* .tran TSTEP TSTOP TSTART TMAX ?UIC?
.OPTIONS GMIN=1e-12 METHOD=trap ABSTOL=1e-12 TEMP=27 srcsteps = 1 gminsteps = 1
*=====
VT      Vtime 0      PWL   ( 0 0 1 1 )
VPM     Vtri  0      PWL   ( 0 0 25m 1 75m -1 100m 0 125m .5 175m -.5 200m 0)
Vsq     Vsq   0      PULSE (.7 1.3 25m 1n 1n 50m 100m )
B1      OUT   0      V =   sin(6.283185307179586232*100*v(Vtime) + 6.283185307179586232*v(Vtri))
B2      OUT2  0      V =   sin(6.283185307179586232*100*v(Vtime)*v(Vsq) )
.tran   .1m   1      0      .1m
```

.control

```
set pensize = 2
run
plot v(out) v(vsq) v(vtri) xlimit 0 200m
plot v(out2) v(vsq) xlimit 0 200m
set pensize = 1
.endc
.end
```


=====**END_OF_SPICE**=====

To Covert PDF to plain text [click below](http://www.fileformat.info/convert/doc/pdf2txt.htm)

<http://www.fileformat.info/convert/doc/pdf2txt.htm>

Suppose you wish to FM a signal like a VCO.
If you simple change the frequency of a sine
wave in an equation as such..

```
VT      Vtime 0      PWL   ( 0 0 1 1 )
Vsq     Vsq   0      PULSE (.7 1.3 25m 1n 1n 50m 100m )
B2      OUT2  0      V =   sin(6.283185307179586232*100*v(Vtime)*v(Vsq) )
```




You will get a discontinuity in phase. But the FM can be done using Phase Modulation by first integrating the modulation signal.

```

VT      Vtime 0      PWL   ( 0 0 1 1 )
VPM     Vtri  0      PWL   ( 0 0 25m 1 75m -1 100m 0 125m .5 175m -.5 200m 0)
B1      OUT   0      V =   sin(6.283185307179586232*100*v(Vtime) + 6.283185307179586232*v(Vtri))

```

For instance, the integral of a square wave is a triangle wave. In this example, two full squares of two different sizes are pre-integrated.



By Phase modulating with such a signal, it is obvious that the frequency modulation tracks the slope (or differential) of the input modulation signal. The PM signal has to FM the carrier first in order to PM the carrier.

So if you wish to generate a FM signal like a VCO, one way to do it in Spice is to first integrate the modulation signal and then apply it to a Phase Modulation equation. PM will automatically differentiates the pre-integrated modulation signal back to its original form before is uses it to FM the carrier.

This method was applied in reverse to build a universal AM Stereo decoder IC. Phase detection can be done using a FM decoder and then integrating the results. This enabled one decoder IC to be able to decode all of the five proposals.